
Some Perturbation Problems

02-08-17

N. T. Gladd

Initialization: Be sure the files *NTGStylesheet2.nb* and *NTGUtilityFunctions.m* is are in the same directory as that from which this notebook was loaded. Then execute the cell immediately below by mousing left on the cell bar to the right of that cell and then typing “shift” + “enter”. Respond “Yes” in response to the query to evaluate initialization cells.

In[88]:=

```
SetDirectory[NotebookDirectory[]];
(* set directory where source files are located *)
SetOptions[EvaluationNotebook[], (* load the StyleSheet *)
  StyleDefinitions -> Get["NTGStylesheet2.nb"]];
Get["NTGUtilityFunctions.m"]; (* Load utilities package *)
```

Purpose

Perturbation methods are broadly useful but often require long repetitive calculations that can be facilitated by the use of Mathematica. Moreover, having immediate access to visualization and numerical methods provides guidance when formulating analytical calculations as well as convenience when checking the accuracy and range of validity of results.

In this notebook, I work through some representative perturbation problems related to algebraic equations in detail. Such problems provide a foundation for the broader application of perturbation methods to differential equations, integrals, and more.

The book *Perturbation Methods*, E. J. Hinch, provides a quite readable introduction to perturbation methods — a topic for which there is a vast literature. Most of the problems in this notebook are drawn from Hinch. Other favorite books are *Introduction to Asymptotics and Special Functions*, F. W. J. Olver, *Advanced Mathematical Methods for Scientists and Engineers*, C. M. Bender and S. A. Orszag, *Introduction to the Foundations of Applied Mathematics*, Mark H. Holmes, and the books by A. H. Nayfeh. On a personal note, I attended and much enjoyed lectures given by Frank Olver, when I was a research faculty member at the University of Maryland in the late 1970s.

The problems treated below are well described in the sources from which they are drawn. My intent is not to repeat the existing exposition but to illustrate how the use of Mathematica can facilitate the

solution process.

The sections below treat

- 1 Iteration and expansion methods for a regular perturbation problem
- 2 A singular perturbation problem
- 3 A problem involving expansion in fractional powers of ϵ
- 4 Another method for determining an appropriate expansion
- 5 A problem for which iteration is preferred

1 Iteration and expansion methods for a regular perturbation problem

In Section 1.1 Hinch considers the model equation

In[5]:=

$$\mathbf{w1[1]} = \mathbf{x^2 + \epsilon x - 1 == 0}$$

Out[5]=

$$-1 + x^2 + x \epsilon == 0$$

with $0 < \epsilon \ll 1$. If ϵ multiplied the term with the highest degree (x^2) this would be a singular perturbation problem. Otherwise it is a regular perturbation problem.

Of course, this model equation is immediately solvable

In[6]:=

$$\mathbf{w1[2]} = \mathbf{Solve[w1[1], x]}$$

Out[6]=

$$\left\{ \left\{ x \rightarrow \frac{1}{2} \left(-\epsilon - \sqrt{4 + \epsilon^2} \right) \right\}, \left\{ x \rightarrow \frac{1}{2} \left(-\epsilon + \sqrt{4 + \epsilon^2} \right) \right\} \right\}$$

1a Iteration

Hinch suggests that the following rearrangement is suitable for application of the iteration method

In[10]:=

$$\mathbf{w1a[1]} = \mathbf{Solve[w1[1] /. x^2 \rightarrow x2, x2][[1, 1] /. x2 \rightarrow x^2 // RE}$$

Out[10]=

$$x^2 == 1 - x \epsilon$$

The lowest order solution assumes $\epsilon = 0$.

In[11]:=

$$\mathbf{w1a[2]} = \mathbf{Solve[w1a[1] /. \epsilon \rightarrow 0, x]}$$

Out[11]=

$$\left\{ \left\{ x \rightarrow -1 \right\}, \left\{ x \rightarrow 1 \right\} \right\}$$

I follow Hinch and consider the development of the iteration solution of the positive branch ($x \rightarrow 1$ as $\epsilon \rightarrow 0$)

In[14]:= `w1a[3] = (sqrt[#]) & /@ w1a[1] // PowerExpand`

Out[14]= $x = \sqrt{1 - x \epsilon}$

The iteration scheme is

In[15]:= `w1a[4] = (w1a[3][[1]] /. x -> xn+1) == (w1a[3][[2]] /. x -> xn)`

Out[15]= $x_{1+n} = \sqrt{1 - \epsilon x_n}$

and the starting point is

In[16]:= `w1a[5] = x0 == 1`

Out[16]= $x_0 = 1$

An analogous scheme could be established for the negative branch.

A recursion that implements this iteration scheme is easy to establish.

In[20]:= `Clear[X];
X[0] = 1;
X[n_] := X[n] = sqrt[1 - epsilon X[n - 1]]`

Some comments on this implementation are

- 1) I used X instead of x so that the earlier formulae would not be “contaminated” by establishing an explicit numerical value for x within the notebook environment.
- 2) I used X[n_] instead of X_n because, as one Mathematica guru on StackExchange puts it, “Don’t use subscripts in calculations unless you are an expert.” The reason for this admonition is that subscripted variables in Mathematica are not interpreted as “Symbols” as intuition would dictate. You can get into trouble by not knowing exactly how Subscripts are implemented. Nonetheless, my background in theoretical physics often compels me to use subscript notation.
- 3) The somewhat mysterious line `X[n_] := X[n] = sqrt[1 - epsilon X[n - 1]]` implements “memoization,” a useful technique by which the values of previous calculations are remembered rather than recalculated (significant time savings).

It is not necessary to loop through the various values of n. Asking for a particular value, say X[6] triggers the recursive valuation of all earlier terms.

In[23]:= `X[6]`

Out[23]=
$$\sqrt{\left(1 - \epsilon \sqrt{\left(1 - \epsilon \sqrt{\left(1 - \epsilon \sqrt{\left(1 - \epsilon \sqrt{1 - \sqrt{1 - \epsilon} \epsilon}\right)}\right)}\right)}\right)}$$

So, if I ask for the values associated with X

In[24]:= `?? X`

Global`X

$$X[0] = 1$$

$$X[1] = \sqrt{1 - \epsilon}$$

$$X[2] = \sqrt{1 - \sqrt{1 - \epsilon} \epsilon}$$

$$X[3] = \sqrt{1 - \epsilon \sqrt{1 - \sqrt{1 - \epsilon} \epsilon}}$$

$$X[4] = \sqrt{1 - \epsilon \sqrt{1 - \epsilon \sqrt{1 - \sqrt{1 - \epsilon} \epsilon}}}$$

$$X[5] = \sqrt{\left(1 - \epsilon \sqrt{1 - \epsilon \sqrt{1 - \epsilon \sqrt{1 - \sqrt{1 - \epsilon} \epsilon}}}\right)}$$

$$X[6] = \sqrt{\left(1 - \epsilon \sqrt{\left(1 - \epsilon \sqrt{\left(1 - \epsilon \sqrt{\left(1 - \epsilon \sqrt{1 - \sqrt{1 - \epsilon} \epsilon}\right)}\right)}\right)}\right)}$$

$$X[n_] := X[n] = \sqrt{1 - \epsilon X[n-1]}$$

For future use I generate a list of these results

In[25]=

```
w1a[6] = Table[{n, X[n]}, {n, 0, 6}]
```

Out[25]=

```
{ {0, 1}, {1, Sqrt[1 - \epsilon]}, {2, Sqrt[1 - Sqrt[1 - \epsilon] \epsilon]}, {3, Sqrt[1 - \epsilon Sqrt[1 - Sqrt[1 - \epsilon] \epsilon]]},
  {4, Sqrt[1 - \epsilon Sqrt[1 - \epsilon Sqrt[1 - Sqrt[1 - \epsilon] \epsilon]]]}, {5, Sqrt[1 - \epsilon Sqrt[1 - \epsilon Sqrt[1 - \epsilon Sqrt[1 - Sqrt[1 - \epsilon] \epsilon]]]}},
  {6, Sqrt[1 - \epsilon Sqrt[1 - \epsilon Sqrt[1 - \epsilon Sqrt[1 - \epsilon Sqrt[1 - Sqrt[1 - \epsilon] \epsilon]]]]]} }
```

and pretty-print them with

In[26]=

```
LGrid[Join[{"n", "Xn"}, w1a[6]], "Iteration Method"]
```

Iteration Method

n	X _n
0	1
1	$\sqrt{1-\epsilon}$
2	$\sqrt{1-\sqrt{1-\epsilon}\epsilon}$
3	$\sqrt{1-\epsilon\sqrt{1-\sqrt{1-\epsilon}\epsilon}}$
4	$\sqrt{1-\epsilon\sqrt{1-\epsilon\sqrt{1-\sqrt{1-\epsilon}\epsilon}}}$
5	$\sqrt{\left(1-\epsilon\sqrt{\left(1-\epsilon\sqrt{\left(1-\epsilon\sqrt{1-\sqrt{1-\epsilon}\epsilon}\right)}\right)}\right)}$
6	$\sqrt{\left(1-\epsilon\sqrt{\left(1-\epsilon\sqrt{\left(1-\epsilon\sqrt{\left(1-\epsilon\sqrt{1-\sqrt{1-\epsilon}\epsilon}\right)}\right)}\right)}\right)}$

Out[26]=

The repeated radicals are not particularly informative so I generate power series representations

In[27]=

```
w1a[7] = ({#[[1]], Series[#[[2]], {ϵ, 0, 5}]} & /@ w1a[6]
```

Out[27]=

$$\left\{ \{0, 1\}, \left\{ 1, 1 - \frac{\epsilon}{2} - \frac{\epsilon^2}{8} - \frac{\epsilon^3}{16} - \frac{5\epsilon^4}{128} - \frac{7\epsilon^5}{256} + O[\epsilon]^6 \right\}, \left\{ 2, 1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} + \frac{\epsilon^3}{8} + \frac{11\epsilon^4}{128} + \frac{3\epsilon^5}{64} + O[\epsilon]^6 \right\}, \right.$$

$$\left. \left\{ 3, 1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{9\epsilon^4}{128} - \frac{5\epsilon^5}{64} + O[\epsilon]^6 \right\}, \left\{ 4, 1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{\epsilon^4}{128} + \frac{\epsilon^5}{32} + O[\epsilon]^6 \right\}, \right.$$

$$\left. \left\{ 5, 1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{\epsilon^4}{128} + O[\epsilon]^6 \right\}, \left\{ 6, 1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{\epsilon^4}{128} + O[\epsilon]^6 \right\} \right\}$$

In[28]=

```
LGrid[Join[{"n", "Xn"}, w1a[7]], "Iterative Expansion"]
```

Iterative Expansion

n	X _n
0	1
1	$1 - \frac{\epsilon}{2} - \frac{\epsilon^2}{8} - \frac{\epsilon^3}{16} - \frac{5\epsilon^4}{128} - \frac{7\epsilon^5}{256} + O[\epsilon]^6$
2	$1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} + \frac{\epsilon^3}{8} + \frac{11\epsilon^4}{128} + \frac{3\epsilon^5}{64} + O[\epsilon]^6$
3	$1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{9\epsilon^4}{128} - \frac{5\epsilon^5}{64} + O[\epsilon]^6$
4	$1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{\epsilon^4}{128} + \frac{\epsilon^5}{32} + O[\epsilon]^6$
5	$1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{\epsilon^4}{128} + O[\epsilon]^6$
6	$1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{\epsilon^4}{128} + O[\epsilon]^6$

Out[28]=

This series can be compared against a power series expansion of the exact solution. For the positive branch that is

In[29]:=

w1[2][[2, 1]]

Out[29]=

$$x \rightarrow \frac{1}{2} \left(-\epsilon + \sqrt{4 + \epsilon^2} \right)$$

In[30]:=

w1a[8] = x → Normal@Series[w1[2][[2, 1, 2]], {ϵ, 0, 5}]

Out[30]=

$$x \rightarrow 1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{\epsilon^4}{128}$$

which agrees with the iteration result at order 6.

In[31]:=

w1a[8][[2]] - w1a[7][[-1, 2]]

Out[31]=

$$0[\epsilon]^6$$

I have focused on the calculation. See Hinch for discussion of merits and demerits of the Iteration method.

1b Expansion method

Hinch develops an approximation for the positive branch by assuming the solution is well represented by the expansion

In[32]:=

expansionRule = x → Sum[ϵⁱ x_i, {i, 0, 7}]

Out[32]=

$$x \rightarrow x_0 + \epsilon x_1 + \epsilon^2 x_2 + \epsilon^3 x_3 + \epsilon^4 x_4 + \epsilon^5 x_5 + \epsilon^6 x_6 + \epsilon^7 x_7$$

In[33]:=

w1b[1] = w1[1] /. expansionRule

Out[33]=

$$-1 + \epsilon \left(x_0 + \epsilon x_1 + \epsilon^2 x_2 + \epsilon^3 x_3 + \epsilon^4 x_4 + \epsilon^5 x_5 + \epsilon^6 x_6 + \epsilon^7 x_7 \right) + \left(x_0 + \epsilon x_1 + \epsilon^2 x_2 + \epsilon^3 x_3 + \epsilon^4 x_4 + \epsilon^5 x_5 + \epsilon^6 x_6 + \epsilon^7 x_7 \right)^2 == 0$$

Collect terms at various orders of ϵ

In[34]:=

w1b[2] = CoefficientList[ExpandAll[w1b[1][[1]], ϵ]

Out[34]=

$$\left\{ -1 + x_0^2, x_0 + 2 x_0 x_1, x_1 + x_1^2 + 2 x_0 x_2, x_2 + 2 x_1 x_2 + 2 x_0 x_3, x_3^2 + x_3 + 2 x_1 x_3 + 2 x_0 x_4, 2 x_2 x_3 + x_4 + 2 x_1 x_4 + 2 x_0 x_5, x_4^2 + 2 x_2 x_4 + x_5 + 2 x_1 x_5 + 2 x_0 x_6, 2 x_3 x_4 + 2 x_2 x_5 + x_6 + 2 x_1 x_6 + 2 x_0 x_7, x_5^2 + 2 x_3 x_5 + 2 x_2 x_6 + x_7 + 2 x_1 x_7, 2 x_4 x_5 + 2 x_3 x_6 + 2 x_2 x_7, x_6^2 + 2 x_4 x_6 + 2 x_3 x_7, 2 x_5 x_6 + 2 x_4 x_7, x_7^2 + 2 x_5 x_7, 2 x_6 x_7, x_7^2 \right\}$$

This leads to a sequence of equations

```
In[35]:= w1b[3] = (# == 0) & /@ w1b[2];
w1b[3] // ColumnForm
```

```
Out[36]= -1 + x0^2 == 0
x0 + 2 x0 x1 == 0
x1 + x1^2 + 2 x0 x2 == 0
x2 + 2 x1 x2 + 2 x0 x3 == 0
x2^2 + x3 + 2 x1 x3 + 2 x0 x4 == 0
2 x2 x3 + x4 + 2 x1 x4 + 2 x0 x5 == 0
x3^2 + 2 x2 x4 + x5 + 2 x1 x5 + 2 x0 x6 == 0
2 x3 x4 + 2 x2 x5 + x6 + 2 x1 x6 + 2 x0 x7 == 0
x4^2 + 2 x3 x5 + 2 x2 x6 + x7 + 2 x1 x7 == 0
2 x4 x5 + 2 x3 x6 + 2 x2 x7 == 0
x5^2 + 2 x4 x6 + 2 x3 x7 == 0
2 x5 x6 + 2 x4 x7 == 0
x6^2 + 2 x5 x7 == 0
2 x6 x7 == 0
x7^2 == 0
```

For $n \geq 1$, the equations are linear and easily solved

```
In[37]:= w1b[4] = Table[Solve[w1b[3][[i]], xi-1][[1, 1]], {i, 2, 8}]
```

```
Out[37]= {x1 -> -1/2, x2 -> (-x1 - x1^2)/(2 x0), x3 -> (-x2 - 2 x1 x2)/(2 x0), x4 -> (-x2^2 - x3 - 2 x1 x3)/(2 x0), x5 -> (-2 x2 x3 - x4 - 2 x1 x4)/(2 x0),
x6 -> (-x3^2 - 2 x2 x4 - x5 - 2 x1 x5)/(2 x0), x7 -> 1/(2 x0) (-2 x3 x4 - 2 x2 x5 - x6 - 2 x1 x6)}
```

For the positive branch $x_0 = 1$

```
In[38]:= w1b[5] = Join[{x0 -> 1}, w1b[4]]
```

```
Out[38]= {x0 -> 1, x1 -> -1/2, x2 -> (-x1 - x1^2)/(2 x0), x3 -> (-x2 - 2 x1 x2)/(2 x0),
x4 -> (-x2^2 - x3 - 2 x1 x3)/(2 x0), x5 -> (-2 x2 x3 - x4 - 2 x1 x4)/(2 x0),
x6 -> (-x3^2 - 2 x2 x4 - x5 - 2 x1 x5)/(2 x0), x7 -> 1/(2 x0) (-2 x3 x4 - 2 x2 x5 - x6 - 2 x1 x6)}
```

and

```
In[39]:= w1b[6] = Join[w1b[5][[1 ;; 2]], Table[w1b[5][[i]] /. w1b[5][[1 ;; i - 1]], {i, 3, 8}]]
```

```
Out[39]= {x0 -> 1, x1 -> -1/2, x2 -> 1/8, x3 -> 0, x4 -> -1/128, x5 -> 0, x6 -> 1/1024, x7 -> 0}
```

Mathematica note: ReplaceRepeated //. is required instead of Replace /.

The expansion method solution is

```
w1b[6] = expansionRule /. w1b[6]
```

$$x \rightarrow 1 - \frac{\epsilon}{2} + \frac{\epsilon^2}{8} - \frac{\epsilon^4}{128} + \frac{\epsilon^6}{1024}$$

which agrees with the results from a power series expansion of the exact solution

2 A singular perturbation problem

Hinch 1.2 treats a singular perturbation problem

```
In[40]=
```

```
w2[1] =  $\epsilon x^2 + x - 1 == 0$ 
```

```
Out[40]=
```

```
 $-1 + x + x^2 \epsilon == 0$ 
```


Complications arise because ϵ multiplies the term with highest degree

A quick visualization indicates two roots that become widely separated as ϵ increases. For $\epsilon = 0$ the equation is linear.

```
In[41]=
```

```
Module[{f, roots, points, inset, lab, legends, eVals = {0.05, 0.1}},
  f[x_,  $\epsilon$ _] :=  $\epsilon x^2 + x - 1$ ;
  roots = NSolve[f[x, #], x] & /@ eVals;
  points = Point[{x, 0}] /. roots;
  lab = Stl@StringForm["roots of ``", TraditionalForm[f[x,  $\epsilon$ ]]];
  legends = {StringForm[" $\epsilon = 0$ "],
    StringForm[" $\epsilon = ``$ ", eVals[[1]]], StringForm[" $\epsilon = 0$ ", eVals[[2]]]};
  Plot[{f[x, 0], f[x, eVals[[1]]], f[x, eVals[[2]]]}, {x, -25, 10},
    PlotStyle -> {Directive[Black, Dashed], Black, Blue},
    AxesLabel -> {Stl["x"], Stl["f[x,  $\epsilon$ "]]},
    PlotLegends -> legends, ImageSize -> 300, PlotLabel -> lab,
    Epilog -> {PointSize[0.02], {Black, points[[1]]}, {Blue, points[[2]]}} ]]
```

```
Out[41]=
```



Solving the quadratic, the exact solutions are

In[42]=

`w2[2] = Solve[w2[1], x]`

Out[42]=

$$\left\{ \left\{ x \rightarrow \frac{-1 - \sqrt{1 + 4\epsilon}}{2\epsilon} \right\}, \left\{ x \rightarrow \frac{-1 + \sqrt{1 + 4\epsilon}}{2\epsilon} \right\} \right\}$$

while the solutions for $\epsilon \ll 1$ are

In[43]=

`w2[3] = (#[[1, 1]] → Normal@Series[#[[1, 2]], {ε, 0, 7}]) & /@ w2[2]`

Out[43]=


$$\left\{ \begin{aligned} x &\rightarrow -1 - \frac{1}{\epsilon} + \epsilon - 2\epsilon^2 + 5\epsilon^3 - 14\epsilon^4 + 42\epsilon^5 - 132\epsilon^6 + 429\epsilon^7, \\ x &\rightarrow 1 - \epsilon + 2\epsilon^2 - 5\epsilon^3 + 14\epsilon^4 - 42\epsilon^5 + 132\epsilon^6 - 429\epsilon^7 \end{aligned} \right\}$$

In[44]=

`Module[{},`

```
Plot[{{ $\frac{-1 - \sqrt{1 + 4\epsilon}}{2\epsilon}$ ,  $-1 - \frac{1}{\epsilon} + \epsilon - 2\epsilon^2$ ,  $\frac{-1 + \sqrt{1 + 4\epsilon}}{2\epsilon}$ ,  $1 - \epsilon + 2\epsilon^2$ }, {ε, 0.01, 1},
PlotStyle → {Black, Directive[Black, Dashed], Blue, Directive[Blue, Dashed]},
AxesLabel → {St1["ε"], St1["roots"]}, ImageSize → 300,
PlotLabel → St1["exact and small ε approximation\nof the two roots"]}]
```

Out[44]=



I apply the expansion method for the root at large negative x .

In this case, the assumed form of the expansion is

In[45]=

`expansionRule = x → Sum[εi xi, {i, -1, 7}]`

Out[45]=

$$x \rightarrow \frac{x_{-1}}{\epsilon} + x_0 + \epsilon x_1 + \epsilon^2 x_2 + \epsilon^3 x_3 + \epsilon^4 x_4 + \epsilon^5 x_5 + \epsilon^6 x_6 + \epsilon^7 x_7$$

Choosing the leading exponent was easy since it was known from the expansion of the exact solution. But — suppose we were tasked with developing an expansion solution for a root of a complicated equation for which we had little insight. One brute force way readily available to a Mathematica user would be to generate a numerical solution and fit it to a polynomial in ϵ

In[46]:=

```
Module[{f, rootNumerical},
  f[x_, ε_] := ε x2 + x - 1;
  rootNumerical =
    Table[{ε, FindRoot[f[x, ε] == 0, {x, -100}][[1, 2]]}, {ε, 0.01, 0.1, 0.01}];
  Fit[rootNumerical, {1/ε2, 1/ε, 1, ε, ε2}, ε]]
```

Out[46]=

$$-0.997874 + \frac{1.60693 \times 10^{-7}}{\epsilon^2} - \frac{1.00003}{\epsilon} + 0.938648 \epsilon - 1.17764 \epsilon^2$$

The smallness of the coefficient of ϵ^{-2} indicates the leading term is ϵ^{-1} . Further on, I'll discuss other methods for determining the expansion rule

In[47]:=

```
w2[4] = w2[1] /. expansionRule
```

Out[47]=

$$-1 + \frac{x-1}{\epsilon} + x_0 + \epsilon x_1 + \epsilon^2 x_2 + \epsilon^3 x_3 + \epsilon^4 x_4 + \epsilon^5 x_5 + \epsilon^6 x_6 + \epsilon^7 x_7 + \epsilon \left(\frac{x-1}{\epsilon} + x_0 + \epsilon x_1 + \epsilon^2 x_2 + \epsilon^3 x_3 + \epsilon^4 x_4 + \epsilon^5 x_5 + \epsilon^6 x_6 + \epsilon^7 x_7 \right)^2 = 0$$

The function `CoefficientList` used above does not work when there are negative exponents, so I write my own algorithm for extracting the equations.

In[48]=

```
w2[5] = Module[{expansion, termm1, term0, termp, eqns},
  expansion = ExpandAll[w2[4][[1]]];
  (* Note the technique used for  $\epsilon^{-1}$  *)
  termm1 = Select[expansion, Not@FreeQ[#, Power[ $\epsilon$ , -1]] &];
  term0 = Select[expansion, FreeQ[#,  $\epsilon$ ] &];
  termp = CoefficientList[expansion - termm1 - term0,  $\epsilon$ ][[2 ;; -1]];
  eqns = {termm1 /.  $\epsilon \rightarrow 1$ , term0, termp} // Flatten;
  (# == 0) & /@ eqns];
w2[5] // ColumnForm
```

Out[49]=

```
x-1 + x-12 == 0
-1 + x0 + 2 x-1 x0 == 0
x02 + x1 + 2 x-1 x1 == 0
2 x0 x1 + x2 + 2 x-1 x2 == 0
x12 + 2 x0 x2 + x3 + 2 x-1 x3 == 0
2 x1 x2 + 2 x0 x3 + x4 + 2 x-1 x4 == 0
x22 + 2 x1 x3 + 2 x0 x4 + x5 + 2 x-1 x5 == 0
2 x2 x3 + 2 x1 x4 + 2 x0 x5 + x6 + 2 x-1 x6 == 0
x32 + 2 x2 x4 + 2 x1 x5 + 2 x0 x6 + x7 + 2 x-1 x7 == 0
2 x3 x4 + 2 x2 x5 + 2 x1 x6 + 2 x0 x7 == 0
x42 + 2 x3 x5 + 2 x2 x6 + 2 x1 x7 == 0
2 x4 x5 + 2 x3 x6 + 2 x2 x7 == 0
x52 + 2 x4 x6 + 2 x3 x7 == 0
2 x5 x6 + 2 x4 x7 == 0
x62 + 2 x5 x7 == 0
2 x6 x7 == 0
x72 == 0
```

Easily solved

In[50]=

```
w2[6] = Table[Solve[w2[5][[i]], xi-2][[1, 1]], {i, 1, 9}]
```

Out[50]=

$$\left\{ \begin{aligned} x_{-1} &\rightarrow -1, x_0 \rightarrow \frac{1}{1 + 2 x_{-1}}, x_1 \rightarrow -\frac{x_0^2}{1 + 2 x_{-1}}, x_2 \rightarrow -\frac{2 x_0 x_1}{1 + 2 x_{-1}}, \\ x_3 &\rightarrow \frac{-x_1^2 - 2 x_0 x_2}{1 + 2 x_{-1}}, x_4 \rightarrow -\frac{2 (x_1 x_2 + x_0 x_3)}{1 + 2 x_{-1}}, x_5 \rightarrow \frac{-x_2^2 - 2 x_1 x_3 - 2 x_0 x_4}{1 + 2 x_{-1}}, \\ x_6 &\rightarrow -\frac{2 (x_2 x_3 + x_1 x_4 + x_0 x_5)}{1 + 2 x_{-1}}, x_7 \rightarrow \frac{-x_3^2 - 2 x_2 x_4 - 2 x_1 x_5 - 2 x_0 x_6}{1 + 2 x_{-1}} \end{aligned} \right\}$$

The explicit coefficients are

In[51]=

```
w2[7] = Join[{w2[6][[1]]}, Table[w2[6][[i]] /. w2[6][[1 ;; i - 1]], {i, 2, 9}]]
```

Out[51]=

```
{x-1 → -1, x0 → -1, x1 → 1, x2 → -2, x3 → 5, x4 → -14, x5 → 42, x6 → -132, x7 → 429}
```

In[52]:= `w2[8] = expansionRule /. w2[7]`

Out[52]=
$$x \rightarrow -1 - \frac{1}{\epsilon} + \epsilon - 2\epsilon^2 + 5\epsilon^3 - 14\epsilon^4 + 42\epsilon^5 - 132\epsilon^6 + 429\epsilon^7$$

Recall the result of the expansion of the exact solution

In[53]:= `w2[3][[1]]`

Out[53]=
$$x \rightarrow -1 - \frac{1}{\epsilon} + \epsilon - 2\epsilon^2 + 5\epsilon^3 - 14\epsilon^4 + 42\epsilon^5 - 132\epsilon^6 + 429\epsilon^7$$

which is identical to the series just calculated

In[54]:= `w2[8][[2]] - w2[3][[1, 2]]`

Out[54]=
$$0$$

Of course, if one found oneself performing repetitive calculations of this sort, it would be easy enough to collect the required operations into a function that could be called for each new calculation.

3 A problem involving expansion in fractional powers of ϵ

In Section 1.3, Hinch discusses an example that requires an expansion in terms of fractional power of ϵ

In[56]:= `w3[1] = (1 - \epsilon) x^2 - 2 x + 1 == 0`

Out[56]=
$$1 - 2x + x^2(1 - \epsilon) == 0$$

I first take advantage of the available exact solution to generate the series that will be later calculated

In[57]:= `w3[2] = Solve[w3[1], x]`

Out[57]=
$$\left\{ \left\{ x \rightarrow \frac{-1 - \sqrt{\epsilon}}{-1 + \epsilon} \right\}, \left\{ x \rightarrow \frac{-1 + \sqrt{\epsilon}}{-1 + \epsilon} \right\} \right\}$$

In[58]:= `w3[3] = w3[2] /. \epsilon \rightarrow 0`

Out[58]=
$$\{ \{ x \rightarrow 1 \}, \{ x \rightarrow 1 \} \}$$

In this case, the expansion involves fractional powers.

In[59]:= `w3[3] = (#[[1, 1]] \to Normal@Series[#[[1, 2]], {\epsilon, 0, 3}]) & /@ w3[2]`

Out[59]=
$$\left\{ x \rightarrow 1 + \sqrt{\epsilon} + \epsilon + \epsilon^{3/2} + \epsilon^2 + \epsilon^{5/2} + \epsilon^3, x \rightarrow 1 - \sqrt{\epsilon} + \epsilon - \epsilon^{3/2} + \epsilon^2 - \epsilon^{5/2} + \epsilon^3 \right\}$$


A visualization provides some insight — there are double roots at $x = 1$ for $\epsilon = 0$, but these roots sepa-

rate for finite ϵ

In[60]=

```
Module[{f},
  f[x_, ε_] := (1 - ε) x^2 - 2 x + 1;
  Plot[{f[x, 0], f[x, 0.1], f[x, 0.2]}, {x, 0, 2},
    PlotStyle -> {Black, Blue, Green}, AxesLabel -> {St1["x"], St1["f[x, ε]"]},
    PlotLegends -> {"ε = 0", "ε = 0.01", "ε = 0.1"}, ImageSize -> 300]
```

Out[60]=



One obstacle when using the expansion method is that a reasonable guess must be made for the form of the expansion. In this case the explicit form is known from the exact solution. But, suppose that we don't have an exact solution. I give two methods for guessing the form of the expansion.

3a Visualize the roots on a log log plot

Numerically determine the roots and plot the deviation from their $\epsilon = 0$ value. In the following I consider the branch for which the roots are less than 1 as ϵ increases, and plot $\log[1 - \text{root}]$ as a function of $\log[\epsilon]$ for small ϵ .


In[61]=

```
Module[{f, rootGuess, roots, rootDiffs, points},
  f[x_, ε_] := (1 - ε) x^2 - 2 x + 1;

  rootGuess = 0.5;
  roots =
    Table[{ε, FindRoot[f[x, ε], {x, rootGuess}][[1, 2]]}, {ε, 0.001, 0.01, 0.001}];
  rootDiffs = ({Log[#[[1]]], Log[1 - #[[2]]]}) & /@ roots;
  points = OC[#, Black] & /@ rootDiffs;

  LogLogPlot[{ε1/4, ε1/3, ε1/2, ε}, {ε, 0.001, 0.01},
    Epilog → points, PlotLegends → {"ε1/4", "ε1/3", "ε1/2", "ε"},
    AxesLabel → {St1["Log[ε]"], St1["Log[1 - f(ε)]"]}]]
```

Out[61]=



From this figure I surmise the expansion should be

$$a_0 + a_1 \epsilon^{\frac{1}{2}} + a_1 \epsilon + \dots$$

A beginning student might reasonably ask the question - Why expend so much effort developing a perturbation solution when a numerical solution is easily obtained?" The answer is that analytical solutions are **extraordinarily** useful in science and mathematics. They allow one to reason about mechanisms, to generalize to other problems, to postpone the use of numerical methods to later stages of analysis, and much more. As Richard Hamming said, "the purpose of computing is insight, not numbers."

3b Fit the deviation of roots from their $\epsilon = 0$ to a functional form

Numerically determine the roots and plot the deviation from their $\epsilon = 0$ value. In the following I consider the branch for which the roots are less than 1 as ϵ increases. I assume that $1 - \text{root}(\epsilon) = a \epsilon^b$ and use the function `NonlinearModelFit` to estimate a and b

In[90]=

```
Module[{f, rootGuess, roots, rootDiffs, points},
  f[x_, ε_] := (1 - ε) x^2 - 2 x + 1;

  rootGuess = 0.5;
  roots =
    Table[{ε, FindRoot[f[x, ε], {x, rootGuess}][[1, 2]]}, {ε, 0.001, 0.01, 0.001}];
  rootDiffs = ({#[[1]], 1 - #[[2]]) & /@ roots;

  NonlinearModelFit[rootDiffs, a ε^b, {a, b}, ε]]
```

Out[90]=

```
FittedModel[0.788439 ε^0.468529]
```

From the fitted result $a = 0.788$, $b = 0.468$ I again surmise the expansion should be

$$a_0 + a_1 \epsilon^{\frac{1}{2}} + a_1 \epsilon + \dots$$

3c Application of expansion method

Armed with a good idea of an appropriate form of the expansion I assume

In[63]=

```
expansionRule = x → 1 + Sum[ε1/2 x1/2i, {i, 1, 4}]
```

Out[63]=

```
x → 1 + √ε x1/2 + ε x1 + ε3/2 x3/2 + ε2 x2
```

Then

In[64]=

```
w3c[1] = w3[1] /. expansionRule
```

Out[64]=

```
1 - 2 (1 + √ε x1/2 + ε x1 + ε3/2 x3/2 + ε2 x2) + (1 - ε) (1 + √ε x1/2 + ε x1 + ε3/2 x3/2 + ε2 x2)2 == 0
```

This leads to the equations

```
In[65]:= w3c[2] = Module[{expansion, eqns},
  (*CoefficientList can be used if
  the expansion variable is forced to have integer powers*)
  expansion = ExpandAll[w3c[1][[1]] /.  $\epsilon \rightarrow \eta^2$  // PowerExpand;
  eqns = CoefficientList[expansion,  $\eta$ ];
  (# == 0) & /@ eqns];
w3c[2] // ColumnForm
```

```
Out[66]= True
True
 $-1 + x_1^2 = 0$ 
 $-2 x_1 + 2 x_1 x_1 = 0$ 
 $-x_1^2 - 2 x_1 + x_1^2 + 2 x_1 x_3 = 0$ 
 $-2 x_1 x_1 - 2 x_3 + 2 x_1 x_3 + 2 x_1 x_2 = 0$ 
 $-x_1^2 - 2 x_1 x_3 + x_3^2 - 2 x_2 + 2 x_1 x_2 = 0$ 
 $-2 x_1 x_3 - 2 x_1 x_2 + 2 x_3 x_2 = 0$ 
 $-x_3^2 - 2 x_1 x_2 + x_2^2 = 0$ 
 $-2 x_3 x_2 = 0$ 
 $-x_2^2 = 0$ 
```

The two lowest order equations in ϵ^0 and $\epsilon^{\frac{1}{2}}$ provide no information

The equation in ϵ provides two values for $x_{\frac{1}{2}}$!

```
In[67]:= w3c[3] = Solve[w3c[2][[3]], x_1/2]
```

```
Out[67]= {{x_1/2 -> -1}, {x_1/2 -> 1}}
```

No multiplicity at order $\epsilon^{\frac{3}{2}}$

```
In[68]:= w3c[4] = Solve[w3c[2][[4]], x_1][[1, 1]]
```

```
Out[68]= x_1 -> 1
```

Proceeding

```
In[69]:= w3c[5] = Solve[w3c[2][[5]], x_3/2][[1, 1]]
```

```
Out[69]= x_3/2 ->  $\frac{x_1^2 + 2 x_1 - x_1^2}{2 x_1}$ 
```


In[70]:= `w3c[6] = Solve[w3c[2][[6]], x2][[1, 1]]`

Out[70]=
$$x_2 \rightarrow \frac{\frac{x_1}{2} x_1 + \frac{x_3}{2} - \frac{x_1}{2} \frac{x_3}{2}}{\frac{x_1}{2}}$$

The general expansion is

In[71]:= `w3c[7] = expansionRule /. w3c[6] /. w3c[5] /. w3c[4]`

Out[71]=
$$x \rightarrow 1 + \epsilon + \epsilon^2 + \sqrt{\epsilon} x_{\frac{1}{2}} + \frac{\epsilon^{3/2} \left(1 + x_{\frac{1}{2}}^2\right)}{2 x_{\frac{1}{2}}}$$

and the specific forms are

In[72]:= `w3c[8] = {w3c[7] /. w3c[3][[1]], w3c[7] /. w3c[3][[2]]}`

Out[72]=
$$\{x \rightarrow 1 - \sqrt{\epsilon} + \epsilon - \epsilon^{3/2} + \epsilon^2, x \rightarrow 1 + \sqrt{\epsilon} + \epsilon + \epsilon^{3/2} + \epsilon^2\}$$

in agreement with the expansion of the exact solutions.

I have explicitly shown all the steps in this example. The various sequential operations could be collected into a single function if there were sufficient motivation.

4 Another method for determining an appropriate expansion

Departing from examples in Hinch I turn to Section 2.2 of *Introduction to the Foundations of Applied Mathematics*, Mark H. Holmes. Holmes discusses a general method for determining expansions and applies it to different example problems. I implement this method in Mathematica for a problem similar to those treated in previous sections.

Consider the regular perturbation problem

In[91]:= `w4[1] = x2 + 2 ε x - 1 == 0`

Out[91]=
$$-1 + x^2 + 2 x \epsilon == 0$$

Holmes suggests the general expansion form

$$x \rightarrow x_0 + x_1 \epsilon^\alpha + x_2 \epsilon^\beta + x_3 \epsilon^\gamma + \dots$$

where it is assumed $\alpha \ll \beta \ll \gamma \ll \dots$ so that $\epsilon^\alpha \gg \epsilon^\beta \gg \epsilon^\gamma \gg \dots$

I will consider only leading terms by explicitly choosing

In[92]:= **expansionRule** = x → x₀ + ε^α x₁ + ε^β x₂

Out[92]= x → x₀ + ε^α x₁ + ε^β x₂

In[93]:= **w4[2]** = w4[1] /. expansionRule // ExpandAll

Out[93]= -1 + 2 ε x₀ + x₀² + 2 ε^{1+α} x₁ + 2 ε^α x₀ x₁ + ε^{2α} x₁² + 2 ε^{1+β} x₂ + 2 ε^β x₀ x₂ + 2 ε^{α+β} x₁ x₂ + ε^{2β} x₂² == 0

At lowest order

In[94]:= **w4[3]** = w4[2] /. ε → 0 /. 0- → 0

Out[94]= -1 + x₀² == 0

there are two roots

In[95]:= **w4[4]** = Solve[w4[3], x₀]

Out[95]= {{x₀ → -1}, {x₀ → 1}}

Normally I would substitute one of these values for x₀ into the perturbation expression and calculate the next order correction. In this case I just substitute the zero order equation and express the higher order corrections in terms of x₀.

In[96]:= **w4[5]** = w4[2] /. x₀² → 1

Out[96]= 2 ε x₀ + 2 ε^{1+α} x₁ + 2 ε^α x₀ x₁ + ε^{2α} x₁² + 2 ε^{1+β} x₂ + 2 ε^β x₀ x₂ + 2 ε^{α+β} x₁ x₂ + ε^{2β} x₂² == 0

At this stage, the lowest order term that remains is linear in ε, the plan is to determine the value of α that will allow that term to be balanced.

I note that the term linear in ε can be selected from the general expression with the operation

In[97]:= **Select**[w4[5][[1]], (MatchQ[#, a_. ε]) &]

Out[97]= 2 ε x₀

and those terms containing ε^α with

In[98]:= **Select**[w4[5][[1]], (Not@FreeQ[#, ε^α]) &]

Out[98]= 2 ε^α x₀ x₁

or with

```
In[99]:= Select[w4[5][[1]], (MatchQ[#, a_.  $\epsilon^\alpha$ ]) &]
```

```
Out[99]:=  $2 \epsilon^\alpha x_0 x_1$ 
```

Thus

```
In[100]:= w4[6] = Select[w4[5][[1]], (Or[MatchQ[#, a_.  $\epsilon$ ], MatchQ[#, a_.  $\epsilon^\alpha$ ]]) &] == 0
```

```
Out[100]:=  $2 \epsilon x_0 + 2 \epsilon^\alpha x_0 x_1 == 0$ 
```

All other terms, such as $\epsilon^{\alpha+1}$ or ϵ^β will be much smaller

From this equation, both α and x_1 must be determined. I'll illustrate an algorithm that accomplishes these tasks. I first proceed step by step and then encapsulate the operations into a general function

An equation for α can be obtained with

```
In[101]:= aside[1] = (w4[6][[1]] /. a_  $\epsilon$  + b_  $\epsilon^\alpha$   $\rightarrow$   $\epsilon - \epsilon^\alpha$ ) == 0
```

```
Out[101]:=  $\epsilon - \epsilon^\alpha == 0$ 
```

and solved

```
In[102]:= aside[2] = Quiet@Solve[aside[1],  $\alpha$ ][[1, 1]]
```

```
Out[102]:=  $\alpha \rightarrow 1$ 
```

Then,

```
In[103]:= aside[3] = {aside[2], Solve[w4[6] /. aside[2],  $x_1$ ][[1, 1]]}
```

```
Out[103]:=  $\{\alpha \rightarrow 1, x_1 \rightarrow -1\}$ 
```

and the order ϵ correction has been obtained.

I define the function

```
In[104]:= Clear[BalanceAtOrder];
BalanceAtOrder[eqn_,  $\epsilon$ Order_, orderVariable_, perturbationVariable_] :=
Module[{orderVariableValue, perturbationVariableValue},
orderVariableValue = Quiet@Solve[ $\epsilon$ Order -  $\epsilon^{\text{orderVariable}} == 0$ , orderVariable][[1, 1]];
perturbationVariableValue =
Solve[(eqn /. orderVariableValue), perturbationVariable][[1, 1]];
{orderVariableValue, perturbationVariableValue}]
```

Using this function, the order ϵ balancing terms are obtain immediately

```
In[106]:= w4[7] = BalanceAtOrder[w4[6],  $\epsilon$ ,  $\alpha$ ,  $x_1$ ]
```

```
Out[106]:=  $\{\alpha \rightarrow 1, x_1 \rightarrow -1\}$ 
```

Proceeding to the next order

In[110]:=

```
w4[8] = w4[5] /. w4[7]
```

Out[110]=

$$-\epsilon^2 + 2 \epsilon^\beta x_0 x_2 + \epsilon^{2\beta} x_2^2 == 0$$

Select the terms that must balance

In[113]:=

```
w4[9] = Select[w4[8][[1]], (Or[MatchQ[#, a_.\epsilon^2], MatchQ[#, a_.\epsilon^\beta]]) &] == 0
```

Out[113]=

$$-\epsilon^2 + 2 \epsilon^\beta x_0 x_2 == 0$$

Then

In[114]:=

```
w4[10] = BalanceAtOrder[w4[9], \epsilon^2, \beta, x2] // PowerExpand
```

Out[114]=

$$\left\{ \beta \rightarrow 2, x_2 \rightarrow \frac{1}{2 x_0} \right\}$$

So, the expansionRule can be made specific

In[115]:=

```
w4[11] = expansionRule /. w4[7] /. w4[10]
```

Out[115]=

$$x \rightarrow -\epsilon + \frac{\epsilon^2}{2 x_0} + x_0$$

I compare this result with the solution of the original equation. The original equation was quadratic and so the exact solution is available. But — suppose the exact solution was not available. I illustrate a numerical approach.

In[116]:=

```
NUMERICALSOLUTION = Flatten /@ Table[{ \epsilon, x /. NSolve[w4[1], x] }, { \epsilon, 0, 1, 0.1}]
```

Out[116]=

```
{{0., -1., 1.}, {0.1, -1.10499, 0.904988}, {0.2, -1.2198, 0.819804},
{0.3, -1.34403, 0.744031}, {0.4, -1.47703, 0.677033},
{0.5, -1.61803, 0.618034}, {0.6, -1.76619, 0.56619}, {0.7, -1.92066, 0.520656},
{0.8, -2.08062, 0.480625}, {0.9, -2.24536, 0.445362}, {1., -2.41421, 0.414214}}
```

where, in this case the equation is a polynomial and NSolve can be used. If the equation was transcendental, I would use FindRoot.

The perturbation solution is

In[117]:=

```
Clear[RootPerturbation];
```


$$\text{RootPerturbation}[\epsilon_, x0_] := -\epsilon + \frac{\epsilon^2}{2 x_0} + x_0$$

Then

In[119]:=

```
Module[{root, εMax, x0, lab},
  lab = Stl@StringForm[
    "solutions of ``\nnumerical (solid) 2nd order perturbation (dashed)",
    TraditionalForm[w4[1]]];
  εMax = NUMERICALSOLUTION[[-1, 1]];
  root[1] = Interpolation@NUMERICALSOLUTION[All, {1, 2}];
  root[2] = Interpolation@NUMERICALSOLUTION[All, {1, 3}];
  x0[1] = 1; x0[2] = -1;
  Plot[{root[1][ε], RootPerturbation[ε, x0[1]],
    root[2][ε], RootPerturbation[ε, x0[2]]}, {ε, 0, εMax},
    PlotStyle → {Black, Directive[Black, Dashed]}, AxesLabel → {Stl[ε], "root"},
    PlotLabel → lab]]
```

Out[119]=



Agreement is quite good, even for ϵ of order 1.

5 A problem for which iteration is preferred

Hinch 1.4 considers the transcendental equation

In[120]:=

```
w5[1] = x Exp[-x] == ε
```

Out[120]=


```
e-x x == ε
```

I visualize the location of roots using

```

In[121]:= Module[{f, root, line, points},
  f[x_, ε_] := x Exp[-x] - ε;
  root[1] = With[{xGuess = 0.01}, FindRoot[f[x, 0.01] == 0, {x, xGuess}][[1, 2]];
  root[2] = With[{xGuess = 5}, FindRoot[f[x, 0.01] == 0, {x, xGuess}][[1, 2]];
  points =
    {PointSize[0.02], Red, Point[{Log@root[1], 0}], Point[{Log@root[2], 0]}};
  line = {Directive[Black, Dashed], Line[{{-Log[0.01], 0}, {-Log[0.01], 1}}]};
  LogLinearPlot[{f[x, 0], f[x, 0.01]}, {x, 0, 8},
    PlotStyle → {Black, Blue}, Epilog → {line, points},
    AxesLabel → {St1["x"], St1["f[x, ε]"]}, PlotLegends → {"ε = 0", "ε = 0.01"} ] ]

```



Hinch suggests an iteration scheme appropriate for the root at large x

```

In[122]:= w5[2] = MapEqn[(# / x) &, w5[1]]

```

Out[122]=

$$e^{-x} \approx \frac{\epsilon}{x}$$

```

In[123]:= w5[3] = MapEqn[(-Log[#]) &, w5[2]] // PowerExpand

```

Out[123]=

$$x \approx \text{Log}[x] - \text{Log}[\epsilon]$$

```

In[124]:= w5[4] = (w5[3][[1]] /. x → x_{n+1}) == (w5[3][[2]] /. x → x_n)

```

Out[124]=

$$x_{1+n} \approx -\text{Log}[\epsilon] + \text{Log}[x_n]$$

The lowest order approximation is

```

In[125]:= w5[5] = x_0 == -Log[ε]

```

Out[125]=

$$x_0 \approx -\text{Log}[\epsilon]$$

To first order

```
In[126]:= w5[6] = w5[4] /. n -> 0 /. (w5[5] // ER)
```

```
Out[126]:= x1 == -Log[ε] + Log[-Log[ε]]
```

and 2nd order


```
In[127]:= w5[7] = w5[4] /. n -> 1 /. (w5[6] // ER)
```

```
Out[127]:= x2 == -Log[ε] + Log[-Log[ε] + Log[-Log[ε]]]
```

Hinch makes the point that this would not have been an easy expansion to guess.

```
In[128]:= Clear[RootOrder1, RootOrder2];
RootOrder1[ε_] := -Log[ε] + Log[-Log[ε]];
RootOrder2[ε_] := -Log[ε] + Log[-Log[ε] + Log[-Log[ε]]];
```

```
In[131]:= Module[{f, fRoot, rootNumerical},
  f[x_, ε_] := x Exp[-x] - ε;
  fRoot[ε_] := FindRoot[f[x, ε] == 0, {x, -Log[ε]}][[1, 2]];
  rootNumerical = Interpolation@Table[{ε, fRoot[ε]}, {ε, 0.1, 0.001, -0.001}];
  Plot[{rootNumerical[ε], RootOrder1[ε], RootOrder2[ε]},
    {ε, 0.001, 0.1}, PlotStyle -> {Black, Blue, Green},
    PlotLegends -> {"numerical", "iteration 1", "iteration 2"},
    AxesLabel -> {St1["ε"], St1["xroot"]}]]
```



Hinch also makes the point that quite small values of ϵ are required to achieve high accuracy